

**Example 1: Unconditional Models for Change in Number Match 3 Response Time**  
(complete data, syntax, and output available for SAS, SPSS, and STATA electronically)

These data come from a short-term longitudinal study of 6 observations over 2 weeks for 101 adults age 65–80 years. The goal is to see how performance on this processing speed task (“number match 3”), as measured by response time in milliseconds, declines (improves) over the 6 practice sessions. We will examine both polynomial and piecewise models to describe individual differences in within-person change across sessions.

**SAS Code for Data Manipulation:**

```
* SAS code to import data;
DATA work.example34; SET example.example34;
  * Center time for polynomial models;
  clsess = Session - 1; LABEL clsess = "clsess: Session Centered at 1";
  * Create piecewise slopes;
  IF Session = 1 THEN DO; Slope12 = 0; Slope26 = 0; END;
  ELSE IF Session = 2 THEN DO; Slope12 = 1; Slope26 = 0; END;
  ELSE IF Session = 3 THEN DO; Slope12 = 1; Slope26 = 1; END;
  ELSE IF Session = 4 THEN DO; Slope12 = 1; Slope26 = 2; END;
  ELSE IF Session = 5 THEN DO; Slope12 = 1; Slope26 = 3; END;
  ELSE IF Session = 6 THEN DO; Slope12 = 1; Slope26 = 4; END;
  LABEL Slope12 = "Slope12: Early Practice Slope (Session 1-2)"
  Slope26 = "Slope26: Later Practice Slope (Session 2-6)"; RUN;
```

**SPSS Code for Data Manipulation:**

```
* SPSS code to import data.
GET FILE = "example/Example34.sav".
DATASET NAME example34 WINDOW=FRONT.
* Center time for polynomial models.
COMPUTE clsess = session - 1.
VARIABLE LABELS clsess "clsess: Session Centered at 1".
* Create piecewise slopes.
RECODE session (1=0) (2 THRU HI=1) INTO Slope12.
RECODE session (LO THRU 2=0) (3=1) (4=2) (5=3) (6=4) INTO Slope26.
VARIABLE LABELS Slope12 "Slope12: Early Practice Slope (Session 1-2)"
Slope26 "Slope26: Later Practice Slope (Session 2-6)".
```

**STATA Code for Data Manipulation:**

```
* Center time for polynomial models (and make quadratic version)
gen clsess = session - 1
label variable clsess "clsess: Session Centered at 1"
gen clsess2 = clsess * clsess
label variable clsess2 "clsess2: Quadratic Session Centered at 1"

* Create piecewise slopes
gen slope12 = session
gen slope26 = session
recode slope12 (1=0) if session==1
recode slope12 (2=1) if session==2
recode slope12 (3=1) if session==3
recode slope12 (4=1) if session==4
recode slope12 (5=1) if session==5
recode slope12 (6=1) if session==6
recode slope26 (1=0) if session==1
recode slope26 (2=0) if session==2
recode slope26 (3=1) if session==3
recode slope26 (4=2) if session==4
recode slope26 (5=3) if session==5
recode slope26 (6=4) if session==6
label variable slope12 "slope12: Early Practice Slope (Session 1-2)"
label variable slope26 "slope26: Later Practice Slope (Session 2-6)"
```



Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr >  Z
UN(1,1)	ID	200883	29471	6.82	<.0001
Session	ID	44900	2825.63	15.89	<.0001

**Calculate the ICC for the Number Match 3 outcome:**

$$ICC = \frac{200883}{200883 + 44900} = .82$$

This null model LRT tells us that the random intercept variance is significantly greater than 0, and thus so is the ICC.

Null Model Likelihood Ratio Test		
DF	Chi-Square	Pr > ChiSq
1	691.74	<.0001

Information Criteria					
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC
8536.9	2	8540.9	8540.9	8543.0	8546.1

REML only counts the # parameters in the model for the variance (not fixed effects).

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	1770.70	45.4206	100	38.98	<.0001

This is the fixed intercept (just the grand mean so far).

**Model 1b. Most Liberal Baseline – Saturated Means, Unstructured Variances (Model Answer Key)**

```
TITLE "SAS Model 1b: Saturated Means, Unstructured Variances";
PROC MIXED DATA=work.example34 NOCLPRINT NOITPRINT COVTEST NAMELEN=100 IC METHOD=REML;
  CLASS ID session;
  MODEL nm3rt = session / SOLUTION DDFM=Satterthwaite;
  REPEATED session / R RCORR TYPE=UN SUBJECT=ID;
  LSMEANS session /; RUN; TITLE1;
```

Placing *session* on the CLASS/BY statements and in the FIXED/MODEL statements treats it as a categorical predictor. So this is an ANOVA means model. No RANDOM statements mean no random effects.

```
TITLE "SPSS Model 1b: Saturated Means, Unstructured Variances".
MIXED nm3rt BY ID session
  /METHOD = REML
  /PRINT = SOLUTION TESTCOV R
  /FIXED = session
  /REPEATED = session | SUBJECT(ID) COVTYPE(UN)
  /EMMEANS = TABLES(session).
```

i. indicates categorical predictor of *session* (ref=last to match others) noconstant = no random intercept (just **R** matrix)

```
* STATA Model 1b: Saturated Means, Unstructured Variances
xtmixed nm3rt ib(last).session, || id: , noconstant ///
  variance reml residuals(unstructured, t(session)),
  estat ic, n(101),
  contrast session, // omnibus test of mean differences
  margins i.session, // observed means per session
  marginsplot name(observed_means, replace) // plot observed means
```

**SAS output:**

Estimated R Matrix for ID 101						
Row	Col1	Col2	Col3	Col4	Col5	Col6
1	<b>301985</b>	235659	217994	202607	192154	195360
2	235659	<b>259150</b>	230217	213232	202092	193268
3	217994	230217	<b>233368</b>	205209	196919	188604
4	202607	213232	205209	<b>217544</b>	193676	185321
5	192154	202092	196919	193676	<b>212098</b>	187840
6	195360	193268	188604	185321	187840	<b>196733</b>

This Unstructured **R** matrix estimates all variances and covariances separately. THIS IS THE DATA we are trying to duplicate with our model for the variance

Estimated R Correlation Matrix for ID 101						
Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.8424	0.8212	0.7905	0.7593	0.8015
2	0.8424	1.0000	0.9361	0.8981	0.8620	0.8559
3	0.8212	0.9361	1.0000	0.9108	0.8851	0.8802
4	0.7905	0.8981	0.9108	1.0000	0.9016	0.8958
5	0.7593	0.8620	0.8851	0.9016	1.0000	0.9196
6	0.8015	0.8559	0.8802	0.8958	0.9196	1.0000

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8229.8	21	8271.8	8273.4	8294.0	8326.7	8347.7

Solution for Fixed Effects

Effect	Session #	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept		1672.14	44.1345	100	37.89	<.0001
Session	1	289.76	32.7000	100	8.86	<.0001
Session	2	143.04	26.2031	100	5.46	<.0001
Session	3	77.8986	22.8842	100	3.40	0.0010
Session	4	45.6604	20.7853	100	2.20	0.0303
Session	5	35.0397	18.1168	100	1.93	0.0559
Session	6	0	.	.	.	.

Mean diffs relative to session 6 (which is the intercept given that it is the highest value)

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
Session	5	100	16.72	<.0001



This is the omnibus test of mean differences across 6 sessions.

Least Squares Means

Effect	Session #	Estimate	Standard Error	DF	t Value	Pr >  t
Session	1	1961.89	54.6805	100	35.88	<.0001
Session	2	1815.17	50.6541	100	35.83	<.0001
Session	3	1750.03	48.0684	100	36.41	<.0001
Session	4	1717.80	46.4101	100	37.01	<.0001
Session	5	1707.18	45.8255	100	37.25	<.0001
Session	6	1672.14	44.1345	100	37.89	<.0001

These are the means per session that the fixed effects will be trying to reproduce.

So here is what are we trying to model—means and variances, where model 1b is the data:



**Model 2a. Fixed Linear Time, Random Intercept**

Level 1:  $y_{ti} = \beta_{0i} + \beta_{1i} (\text{Session}_{ti} - 1) + e_{ti}$   
 Level 2: Intercept:  $\beta_{0i} = \gamma_{00} + U_{0i}$   
 Linear Session:  $\beta_{1i} = \gamma_{10}$

```
TITLE "SAS Model 2a: Fixed Linear Time, Random Intercept";
PROC MIXED DATA=work.example35 NOCLPRINT NOITPRINT COVTEST NAMELEN=100 IC METHOD=REML;
  CLASS ID session;
  MODEL nm3rt = c1sess / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / G V VCORR TYPE=UN SUBJECT=ID;
  REPEATED session / R TYPE=VC SUBJECT=ID; RUN; TITLE1;
```

```
TITLE "SPSS Model 2a: Fixed Linear Time, Random Intercept".
MIXED nm3rt BY ID session WITH c1sess
  /METHOD = REML
  /PRINT = SOLUTION TESTCOV G R
  /FIXED = c1sess
  /RANDOM = INTERCEPT | SUBJECT(ID) COVTYPE(UN)
  /REPEATED = session | SUBJECT(ID) COVTYPE(ID).
```

The predictor of *c1sess* will be treated as continuous given that it is not on the CLASS statement (SAS) and it is on WITH (SPSS).

```
* STATA Model 2a: Fixed Linear Time, Random Intercept
xtmixed nm3rt c.c1sess, || id: , ///
  variance reml covariance(un) residuals(independent,t(session)),
  estat ic, n(101),
  estimates store FixLin
```

DV = nm3rt, c. means continuous fixed slope for *c1sess*  
 Level 2 ID is id, random intercept by default  
 estimates → save results as “FixLin” for next LRT

**SAS output:**

Estimated V Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	238084	202422	202422	202422	202422	202422
2	202422	238084	202422	202422	202422	202422
3	202422	202422	238084	202422	202422	202422
4	202422	202422	202422	238084	202422	202422
5	202422	202422	202422	202422	238084	202422
6	202422	202422	202422	202422	202422	238084

Estimated V Correlation Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.8502	0.8502	0.8502	0.8502	0.8502
2	0.8502	1.0000	0.8502	0.8502	0.8502	0.8502
3	0.8502	0.8502	1.0000	0.8502	0.8502	0.8502
4	0.8502	0.8502	0.8502	1.0000	0.8502	0.8502
5	0.8502	0.8502	0.8502	0.8502	1.0000	0.8502
6	0.8502	0.8502	0.8502	0.8502	0.8502	1.0000

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z	Pr >  Z
UN(1,1)	ID	202422	29470	6.87	<.0001
Session	ID	35662	2246.48	15.87	<.0001

Relative to the empty means, random intercept model 1a, the fixed linear effect of session explained ~21% of the residual variance (which made the random intercept variance increase due to its smaller residual variance correction factor).

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8414.7	2	8418.7	8418.7	8420.8	8423.9	8425.9

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	1899.63	46.7882	113	40.60	<.0001
C1sess	-51.5719	4.4918	504	-11.48	<.0001

The fixed linear effect of *c1sess* is significant according to the Wald test (*p*-value for fixed effect).

**Model 2b. Random Linear Time**

$$\text{Level 1: } y_{ti} = \beta_{0i} + \beta_{1i} (\text{Session}_{ti} - 1) + e_{ti}$$

$$\text{Level 2: Intercept: } \beta_{0i} = \gamma_{00} + U_{0i}$$

$$\text{Linear Session: } \beta_{1i} = \gamma_{10} + U_{1i}$$

```
TITLE "SAS Model 2b: Random Linear Time";
PROC MIXED DATA=work.example35 NOCLPRINT NOITPRINT COVTEST NAMELEN=100 IC METHOD=REML;
  CLASS ID session;
  MODEL nm3rt = c1sess / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT c1sess / G V V CORR TYPE=UN SUBJECT=ID;
  REPEATED session / R TYPE=VC SUBJECT=ID; RUN; TITLE1;
```

```
TITLE "SPSS Model 2b: Random Linear Time".
MIXED nm3rt BY ID session WITH c1sess
  /METHOD = REML
  /PRINT = SOLUTION TESTCOV G R
  /FIXED = c1sess
  /RANDOM = INTERCEPT c1sess | SUBJECT(ID) COVTYPE(UN)
  /REPEATED = session | SUBJECT(ID) COVTYPE(ID).
```

Now there are 2 random effects: intercept and linear slope, given by c1sess on the RANDOM statements.

```
* STATA Model 2b: Random Linear Time
xtmixed nm3rt c.c1sess, || id: c1sess, ///
  variance reml covariance(un) residuals(independent,t(session)),
  estat ic, n(101),
  estat recovariance, level(id),
  estimates store RandLin,
  lrtest RandLin FixLin
```

DV = nm3rt, c. means continuous fixed slope for c1sess  
 Level 2 ID is id, random intercept and c1sess now estimates → save results as “RandLin” for LRT

**SAS output:**

Estimated R Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	27905					
2		27905				
3			27905			
4				27905		
5					27905	
6						27905

Estimated G Matrix Participant

Row	Effect	ID	Col1	Col2
1	Intercept	101	253258	-12701
2	C1sess	101	-12701	2233.83

Estimated G Correlation Matrix ID: Participant

Row	Effect	ID	Col1	Col2
1	Intercept	101	1.0000	-0.5340
2	c1sess	101	-0.5340	1.0000

GCORR shows the correlation among random effects.

Estimated V Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	<b>281163</b>	240557	227856	215155	202455	189754
2	240557	<b>257995</b>	219623	209156	198689	188222
3	227856	219623	<b>239295</b>	203157	194924	186691
4	215155	209156	203157	<b>225063</b>	191158	185159
5	202455	198689	194924	191158	<b>215298</b>	183627
6	189754	188222	186691	185159	183627	<b>210001</b>

The V matrix is the total variance-covariance matrix after combining the level-2 G and level-1 R matrices. Now the variances and covariances are predicted to change based on time.

**How the V matrix variances and covariances get calculated in a random linear time model:**

$$V_i \text{ matrix: Variance}[y_{time}] = \tau_{U_0}^2 + [(\text{Session} - 1)^2 \tau_{U_1}^2] + [2(\text{Session} - 1)\tau_{U_{01}}] + \sigma_e^2$$

$$V_i \text{ matrix: Covariance}[y_A, y_B] = \tau_{U_0}^2 + [(A + B)\tau_{U_{01}}] + [(AB)\tau_{U_1}^2]$$

Estimated V Correlation Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.8932	0.8784	0.8553	0.8229	0.7809
2	0.8932	1.0000	0.8839	0.8680	0.8430	0.8086
3	0.8784	0.8839	1.0000	0.8754	0.8588	0.8328
4	0.8553	0.8680	0.8754	1.0000	0.8684	0.8517
5	0.8229	0.8430	0.8588	0.8684	1.0000	0.8636
6	0.7809	0.8086	0.8328	0.8517	0.8636	1.0000

The **VCORR** matrix is the correlation version. The ICC is now predicted to change over time, too (and conditional on linear time).

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z	Pr >  Z
UN(1,1)	ID	253258	37897	6.68	<.0001
UN(2,1)	ID	-12701	3621.98	-3.51	0.0005
UN(2,2)	ID	2233.83	552.92	4.04	<.0001
Session	ID	27905	1963.42	14.21	<.0001

Is the random linear time model (2b) better than the fixed linear time, random intercept model (2a)?

Yep,  $-2\Delta LL = 43$ , which is bigger than the critical value of 5.99ish on  $df = 2$ ish

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8372.1	4	8380.1	8380.2	8384.3	8390.6	8394.6

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	1899.63	51.4998	100	36.89	<.0001
C1sess	-51.5719	6.1567	100	-8.38	<.0001

**Model 3a. Fixed Quadratic, Random Linear Time**

Level 1:  $y_{ti} = \beta_{0i} + \beta_{1i}(\text{Session}_{ti} - 1) + \beta_{2i}(\text{Session}_{ti} - 1)^2 + e_{ti}$   
 Level 2: Intercept:  $\beta_{0i} = \gamma_{00} + U_{0i}$   
 Linear Session:  $\beta_{1i} = \gamma_{10} + U_{1i}$   
 Quadratic Session:  $\beta_{2i} = \gamma_{20}$

```
TITLE1 "SAS Model 3a: Fixed Quadratic, Random Linear Time";
PROC MIXED DATA=work.example35 NOCLPRINT NOITPRINT COVTEST NAMELEN=100 IC METHOD=REML;
  CLASS ID session;
  MODEL nm3rt = c1sess c1sess*c1sess / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT c1sess / G V VCORR TYPE=UN SUBJECT=ID;
  REPEATED session / R TYPE=VC SUBJECT=ID; RUN; TITLE1;
```

Interactions can be defined on the fly in SAS and SPSS using \*.

```
TITLE "SPSS Model 3a: Fixed Quadratic, Random Linear Time".
MIXED nm3rt BY ID session WITH c1sess
  /METHOD = REML
  /PRINT = SOLUTION TESTCOV G R
  /FIXED = c1sess c1sess*c1sess
  /RANDOM = INTERCEPT c1sess | SUBJECT(ID) COVTYPE(UN)
  /REPEATED = session | SUBJECT(ID) COVTYPE(ID).
```

```
* STATA Model 3a: Fixed Quadratic, Random Linear Time
xtmixed nm3rt c.c1sess c.c1sess#c.c1sess, || id: c1sess, ///
  variance reml covariance(un) residuals(independent,t(session)),
  estat ic, n(101),
```

Interactions can be defined on the fly in STATA using # for fixed effects, but not for random effects.

```
estat recovariance, level(id),
estimates store FixQuad
```

**SAS output:**

Estimated R Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	26176					
2		26176				
3			26176			
4				26176		
5					26176	
6						26176

Estimated G Matrix  
ID:  
Participant

Row	Effect	ID	Col1	Col2
1	Intercept	101	254164	-12948
2	c1sess	101	-12948	2332.67

GCORR shows intercept-slope correlation  $r = -.53$

Estimated V Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	280339	241216	228268	215320	202372	189424
2	241216	256776	219985	209370	198755	188140
3	228268	219985	237879	203420	195138	186855
4	215320	209370	203420	223646	191521	185571
5	202372	198755	195138	191521	214079	184286
6	189424	188140	186855	185571	184286	209178

Estimated V Correlation Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.8991	0.8839	0.8599	0.8261	0.7822
2	0.8991	1.0000	0.8901	0.8737	0.8477	0.8118
3	0.8839	0.8901	1.0000	0.8819	0.8647	0.8377
4	0.8599	0.8737	0.8819	1.0000	0.8753	0.8580
5	0.8261	0.8477	0.8647	0.8753	1.0000	0.8709
6	0.7822	0.8118	0.8377	0.8580	0.8709	1.0000

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr >  Z
UN(1,1)	ID	254164	37896	6.71	<.0001
UN(2,1)	ID	-12948	3620.70	-3.58	0.0003
UN(2,2)	ID	2332.67	551.58	4.23	<.0001
session	ID	26176	1844.01	14.20	<.0001

Relative to the random linear time model 2b, the fixed quadratic effect of session explained another ~6% of the residual variance (which made the random intercept variance increase due to its residual variance correction factor).

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8341.5	4	8349.5	8349.5	8353.7	8359.9	8363.9

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	1945.85	52.2433	106	37.25	<.0001
C1sess	-120.90	14.5415	502	-8.31	<.0001
C1sess*C1sess	13.8656	2.6348	403	5.26	<.0001

The fixed quadratic effect of *c1sess* is significant according to the Wald test (*p*-value for fixed effect).



**Model 3b. Random Quadratic Time (and an example of ESTIMATE/TEST/LINCOM statements)**

$$\text{Level 1: } y_{ti} = \beta_{0i} + \beta_{1i}(\text{Session}_{ti} - 1) + \beta_{2i}(\text{Session}_{ti} - 1)^2 + e_{ti}$$

$$\text{Level 2: Intercept: } \beta_{0i} = \gamma_{00} + U_{0i}$$

$$\text{Linear Session: } \beta_{1i} = \gamma_{10} + U_{1i}$$

$$\text{Quadratic Session: } \beta_{2i} = \gamma_{20} + U_{2i}$$

```
TITLE1 "SAS Model 3b: Random Quadratic Time";
PROC MIXED DATA=work.example35 NOCLPRINT NOITPRINT COVTEST NAMELEN=100 IC METHOD=REML;
  CLASS ID session;
  MODEL nm3rt = c1sess c1sess*c1sess / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT c1sess c1sess*c1sess / G V VCORR TYPE=UN SUBJECT=ID;
  REPEATED session / R TYPE=VC SUBJECT=ID;
  ESTIMATE "Intercept at Session 1"      intercept 1 c1sess 0      c1sess*c1sess 0;
  ESTIMATE "Intercept at Session 2"      intercept 1 c1sess 1      c1sess*c1sess 1;
  ESTIMATE "Intercept at Session 3"      intercept 1 c1sess 2      c1sess*c1sess 4;
  ESTIMATE "Intercept at Session 4"      intercept 1 c1sess 3      c1sess*c1sess 9;
  ESTIMATE "Intercept at Session 5"      intercept 1 c1sess 4      c1sess*c1sess 16;
  ESTIMATE "Intercept at Session 6"      intercept 1 c1sess 5      c1sess*c1sess 25;
  * Predicting linear rate of change at each session (linear changes by 2*quad);
  ESTIMATE "Linear Slope at Session 1"    c1sess 1      c1sess*c1sess 0;
  ESTIMATE "Linear Slope at Session 2"    c1sess 1      c1sess*c1sess 2;
  ESTIMATE "Linear Slope at Session 3"    c1sess 1      c1sess*c1sess 4;
  ESTIMATE "Linear Slope at Session 4"    c1sess 1      c1sess*c1sess 6;
  ESTIMATE "Linear Slope at Session 5"    c1sess 1      c1sess*c1sess 8;
  ESTIMATE "Linear Slope at Session 6"    c1sess 1      c1sess*c1sess 10; RUN; TITLE1;
```

```
TITLE "SPSS Model 3b: Random Quadratic Time".
MIXED nm3rt BY ID session WITH c1sess
  /METHOD = REML
  /PRINT = SOLUTION TESTCOV G R
  /FIXED = c1sess c1sess*c1sess
  /RANDOM = INTERCEPT c1sess c1sess*c1sess | SUBJECT(ID) COVTYPE(UN)
  /REPEATED = session | SUBJECT(ID) COVTYPE(ID)
  /TEST = "Intercept at Session 1"      intercept 1 c1sess 0      c1sess*c1sess 0
  /TEST = "Intercept at Session 2"      intercept 1 c1sess 1      c1sess*c1sess 1
  /TEST = "Intercept at Session 3"      intercept 1 c1sess 2      c1sess*c1sess 4
  /TEST = "Intercept at Session 4"      intercept 1 c1sess 3      c1sess*c1sess 9
  /TEST = "Intercept at Session 5"      intercept 1 c1sess 4      c1sess*c1sess 16
  /TEST = "Intercept at Session 6"      intercept 1 c1sess 5      c1sess*c1sess 25
  /TEST = "Linear Slope at Session 1"    c1sess 1      c1sess*c1sess 0
  /TEST = "Linear Slope at Session 2"    c1sess 1      c1sess*c1sess 2
  /TEST = "Linear Slope at Session 3"    c1sess 1      c1sess*c1sess 4
  /TEST = "Linear Slope at Session 4"    c1sess 1      c1sess*c1sess 6
  /TEST = "Linear Slope at Session 5"    c1sess 1      c1sess*c1sess 8
  /TEST = "Linear Slope at Session 6"    c1sess 1      c1sess*c1sess 10.
```

```
* STATA Model 3b: Random Quadratic Time
xtmixed nm3rt c.c1sess c.c1sess#c.c1sess, || id: c1sess c1sess2, ///
  variance reml covariance(un) residuals(independent,t(session)),
  estat ic, n(101),
  estat recovariance, level(id),
  estimates store RandQuad,
  lrtest RandQuad FixQuad,
  margins, at(c.c1sess=(0(1)5)) vsquish // intercepts per session
  marginsplot, name(predicted_means, replace) // plot intercepts
  margins, at(c.c1sess=(0(1)5)) dydx(c.c1sess) vsquish // linear slope per session
  marginsplot, name(change_in_linear_slope, replace) // plot quadratic effect
```

The random statement will not accept interaction terms, so we are using the *c1sess2* created manually before.

**SAS output:**

Estimated R Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	20298					
2		20298				
3			20298			
4				20298		
5					20298	
6						20298

Estimated G Matrix  
Participant

Row	Effect	ID	Col1	Col2	Col3
1	Intercept	101	276206	-35734	3901.96
2	C1sess	101	-35734	25840	-3903.32
3	C1sess*C1sess	101	3901.96	-3903.32	634.47

Estimated G Correlation Matrix  
ID:  
Participant

Row	Effect	ID	Col1	Col2	Col3
1	Intercept	101	1.0000	-0.4230	0.2948
2	c1sess	101	-0.4230	1.0000	-0.9640
3	c1sess*c1sess	101	0.2948	-0.9640	1.0000

Estimated V Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	<b>296504</b>	244374	220346	204122	195702	195085
2	244374	<b>251508</b>	219312	208680	199315	191215
3	220346	219312	<b>235842</b>	209043	199808	187840
4	204122	208680	209043	<b>225508</b>	197182	184958
5	195702	199315	199808	197182	<b>211735</b>	182571
6	195085	191215	187840	184958	182571	<b>200977</b>

The **V** matrix is the total variance-covariance matrix after combining the level-2 **G** and level-1 **R** matrices. The variances and covariances are predicted to change based on time, but differently.

**How the V matrix variances and covariances get calculated in a random quadratic time model:**

Predicted Variance at Time *T*:

$$\text{Var}(y_T) = \sigma_e^2 + \tau_{U_0}^2 + 2*T*\tau_{U_{01}} + T^2*\tau_{U_1}^2 + 2*T^2*\tau_{U_{02}} + 2*T^3*\tau_{U_{12}} + T^4*\tau_{U_2}^2$$

Predicted Covariance between Time *A* and *B*:

$$\text{Cov}(y_A, y_B) = \tau_{U_0}^2 + (A+B)*\tau_{U_{01}} + (AB)*\tau_{U_1}^2 + (A^2+B^2)*\tau_{U_{02}} + (AB^2)+(A^2B)*\tau_{U_{12}} + (A^2B^2)*\tau_{U_2}^2$$

Estimated V Correlation Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.8949	0.8333	0.7894	0.7811	0.7992
2	0.8949	1.0000	0.9005	0.8762	0.8637	0.8505
3	0.8333	0.9005	1.0000	0.9064	0.8941	0.8628
4	0.7894	0.8762	0.9064	1.0000	0.9024	0.8688
5	0.7811	0.8637	0.8941	0.9024	1.0000	0.8850
6	0.7992	0.8505	0.8628	0.8688	0.8850	1.0000

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z	Pr >  Z
UN(1,1)	ID	276206	41442	6.66	<.0001
UN(2,1)	ID	-35734	11941	-2.99	0.0028
UN(2,2)	ID	25840	5864.41	4.41	<.0001
UN(3,1)	ID	3901.96	1949.06	2.00	0.0453
UN(3,2)	ID	-3903.32	982.61	-3.97	<.0001
UN(3,3)	ID	634.47	172.37	3.68	0.0001
Session	ID	20298	1649.11	12.31	<.0001

Neg2LogLike	Parms	Information Criteria			BIC	CAIC
		AIC	AICC	HQIC		
8302.7	7	8316.7	8316.9	8324.2	8335.1	8342.1

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	1945.85	53.8497	100	36.13	<.0001
C1sess	-120.90	20.0476	100	-6.03	<.0001
C1sess*C1sess	13.8656	3.4154	100	4.06	<.0001

Is the random quadratic model (3b) better than the fixed quadratic, random linear model (3a)?  
  
 Yep,  $-2\Delta LL = 39$ , which is bigger than the critical value of 7.82ish on  $df \approx 3$ ish

Computing random effects confidence intervals for each random effect:

Random Effect 95% CI = fixed effect  $\pm (1.96 * \sqrt{\text{Random Variance}})$

Intercept 95% CI =  $\gamma_{00} \pm (1.96 * \sqrt{\tau_{U_0}^2}) \rightarrow 1,945.9 \pm (1.96 * \sqrt{276,209}) = 916 \text{ to } 2,976$

Linear Time Slope 95% CI =  $\gamma_{10} \pm (1.96 * \sqrt{\tau_{U_1}^2}) \rightarrow -120.9 \pm (1.96 * \sqrt{25,840}) = -436 \text{ to } 194$

Quadratic Time Slope 95% CI =  $\gamma_{20} \pm (1.96 * \sqrt{\tau_{U_2}^2}) \rightarrow 13.9 \pm (1.96 * \sqrt{634}) = -36 \text{ to } 63$



Estimates

Label	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept at Session 1	1945.85	53.8497	100	36.13	<.0001
Intercept at Session 2	1838.82	48.4864	100	37.92	<.0001
Intercept at Session 3	1759.51	46.9973	100	37.44	<.0001
Intercept at Session 4	1707.94	45.8959	100	37.21	<.0001
Intercept at Session 5	1684.10	44.2395	100	38.07	<.0001
Intercept at Session 6	1687.99	44.2038	100	38.19	<.0001
Linear Trend at Session 1	-120.90	20.0476	100	-6.03	<.0001
Linear Trend at Session 2	-93.1687	13.6497	100	-6.83	<.0001
Linear Trend at Session 3	-65.4375	8.0028	100	-8.18	<.0001
Linear Trend at Session 4	-37.7062	5.9242	100	-6.36	<.0001
Linear Trend at Session 5	-9.9750	9.9733	100	-1.00	0.3196
Linear Trend at Session 6	17.7562	16.0362	100	1.11	0.2708

These are the quadratic-model-predicted means (intercepts) per session.

These are the instantaneous linear slopes at each session. Note how the SEs narrow towards the middle sessions.

How well do the predicted means, variances, and covariances from the random quadratic model (3b) match the original means, variances, and covariances from the saturated means model (1b)?



The quadratic model appears to be a good contender, but let's examine how well a two-slope piecewise model might fit these same data...

**Model 4a: Fixed Slope12, Fixed Slope26, Random Intercept Model**

$$\text{Level 1: } y_{ti} = \beta_{0i} + \beta_{1i}(\text{Slope12}_{ti}) + \beta_{2i}(\text{Slope26}_{ti}) + e_{ti}$$

$$\text{Level 2: Intercept: } \beta_{0i} = \gamma_{00} + U_{0i}$$

$$\text{Slope12: } \beta_{1i} = \gamma_{10}$$

$$\text{Slope26: } \beta_{2i} = \gamma_{20}$$

```
TITLE "Model 4a: Fixed Slope12, Fixed Slope26, Random Intercept Model";
PROC MIXED DATA=work.example34 NOCLPRINT NOITPRINT COVTEST NAMELEN=100 IC METHOD=REML;
  CLASS ID session;
  MODEL nm3rt = Slope12 Slope26 / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / V VCORR TYPE=UN SUBJECT=ID;
  REPEATED session / R TYPE=VC SUBJECT=ID; RUN; TITLE1;
```

```
TITLE "SPSS Model 4a: Fixed Slope12, Fixed Slope26, Random Intercept Model".
MIXED nm3rt BY ID session WITH Slope12 Slope26
  /METHOD = REML
  /PRINT = SOLUTION TESTCOV G R
  /FIXED = Slope12 Slope26
  /RANDOM = INTERCEPT | SUBJECT(ID) COVTYPE(UN)
  /REPEATED = session | SUBJECT(ID) COVTYPE(ID).
```

```
* STATA Model 4a: Fixed Slope12, Fixed Slope26, Random Intercept Model
xtmixed nm3rt c.slope12 c.slope26, || id:, ///
  variance reml covariance(un) residuals(independent,t(session)),
  estat ic, n(101),
  estimates store FixPiece
```

**SAS output:**

Estimated R Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	34098					
2		34098				
3			34098			
4				34098		
5					34098	
6						34098

This **R matrix** VC structure (equal variance over time, no covariance of any kind) will be used repeatedly as we add fixed and random piecewise slopes to the model.

Estimated G Matrix

Row	Effect	Person ID	Col1
1	Intercept	101	202683

Estimated V Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	<b>236781</b>	202683	202683	202683	202683	202683
2	202683	<b>236781</b>	202683	202683	202683	202683
3	202683	202683	<b>236781</b>	202683	202683	202683
4	202683	202683	202683	<b>236781</b>	202683	202683
5	202683	202683	202683	202683	<b>236781</b>	202683
6	202683	202683	202683	202683	202683	<b>236781</b>

This random intercept model predicts a compound symmetry pattern for the **V matrix** (equal variance, equal covariance over time).

Estimated V Correlation Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.8560	0.8560	0.8560	0.8560	0.8560
2	0.8560	1.0000	0.8560	0.8560	0.8560	0.8560

3	0.8560	0.8560	1.0000	0.8560	0.8560	0.8560
4	0.8560	0.8560	0.8560	1.0000	0.8560	0.8560
5	0.8560	0.8560	0.8560	0.8560	1.0000	0.8560
6	0.8560	0.8560	0.8560	0.8560	0.8560	1.0000

Covariance Parameter Estimates


Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr >  Z
UN(1,1)	ID	202683	29470	6.88	<.0001
Session	ID	34098	2150.11	15.86	<.0001

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8382.7	2	8386.7	8386.7	8388.8	8391.9	8393.9

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t	
Intercept	1961.89	48.4187	129	40.52	<.0001	
Slope12	-163.64	23.2415	503	-7.04	<.0001	RATE OF CHANGE FROM SESSION 1-2
Slope26	-32.8932	5.8104	503	-5.66	<.0001	RATE OF CHANGE FROM SESSION 2-6



**Model 4b: Random Slope12, Fixed Slope26 Model**

Level 1:  $y_{ti} = \beta_{0i} + \beta_{1i}(\text{Slope12}_{ti}) + \beta_{2i}(\text{Slope26}_{ti}) + e_{ti}$   
 Level 2: Intercept:  $\beta_{0i} = \gamma_{00} + U_{0i}$   
 Slope12:  $\beta_{1i} = \gamma_{10} + U_{1i}$   
 Slope26:  $\beta_{2i} = \gamma_{20}$

```
TITLE "Model 4b: Random Slope12, Fixed Slope26 Model";
PROC MIXED DATA=work.example34 NOCLPRINT NOITPRINT COVTEST NAMELEN=100 IC METHOD=REML;
  CLASS ID session;
  MODEL nm3rt = Slope12 Slope26 / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT Slope12 / G GCORR V VCORR TYPE=UN SUBJECT=ID;
  REPEATED session / R TYPE=VC SUBJECT=ID; RUN; TITLE1;
TITLE "SPSS Model 4b: Random Slope12, Fixed Slope26 Model".
```

```

MIXED nm3rt BY ID session WITH Slope12 Slope26
  /METHOD = REML
  /PRINT = SOLUTION TESTCOV G R
  /FIXED = Slope12 Slope26
  /RANDOM = INTERCEPT Slope12 | SUBJECT(ID) COVTYPE(UN)
  /REPEATED = session | SUBJECT(ID) COVTYPE(ID).

* STATA Model 4b: Random Slope12, Fixed Slope26 Model
xtmixed nm3rt c.slope12 c.slope26, || id: slope12, ///
  variance reml covariance(un) residuals(independent,t(session)),
  estat ic, n(101),
  estat recovariance, level(id),
  estimates store RandP1,
  lrtest RandP1 FixPiece

```

**SAS output:**

Estimated R Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	24168					
2		24168				
3			24168			
4				24168		
5					24168	
6						24168

Estimated G Matrix

Row	Effect	Person ID	Col1	Col2
1	Intercept	101	277818	-69063
2	Slope12	101	-69063	59941

Estimated G Correlation Matrix

Row	Effect	Person ID	Col1	Col2
1	Intercept	101	1.0000	-0.5352
2	Slope12	101	-0.5352	1.0000

Estimated V Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	<b>301985</b>	208755	208755	208755	208755	208755
2	208755	<b>223800</b>	199632	199632	199632	199632
3	208755	199632	<b>223800</b>	199632	199632	199632
4	208755	199632	199632	<b>223800</b>	199632	199632
5	208755	199632	199632	199632	<b>223800</b>	199632
6	208755	199632	199632	199632	199632	<b>223800</b>

The pattern of variance and covariances in **V** is now compound symmetry from session 2 onward because slope26 is fixed.

Estimated V Correlation Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.8030	0.8030	0.8030	0.8030	0.8030
2	0.8030	1.0000	0.8920	0.8920	0.8920	0.8920
3	0.8030	0.8920	1.0000	0.8920	0.8920	0.8920
4	0.8030	0.8920	0.8920	1.0000	0.8920	0.8920
5	0.8030	0.8920	0.8920	0.8920	1.0000	0.8920
6	0.8030	0.8920	0.8920	0.8920	0.8920	1.0000

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z	Pr >  Z
UN(1,1)	ID	277818	42741	6.50	<.0001
UN(2,1)	ID	-69063	18932	-3.65	0.0003
UN(2,2)	ID	59941	12743	4.70	<.0001
Session	ID	24168	1702.53	14.20	<.0001

Neg2LogLike	Parms	Information Criteria			BIC	CAIC
		AIC	AICC	HQIC		
8319.6	4	8327.6	8327.7	8331.8	8338.1	8342.1

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	1961.89	54.6805	100	35.88	<.0001
Slope12	-163.64	31.2462	123	-5.24	<.0001 RATE OF CHANGE FROM SESSION 1-2
Slope26	-32.8932	4.8916	403	-6.72	<.0001 RATE OF CHANGE FROM SESSION 2-6

Is random slope12 significant?  
 Yep,  $-2\Delta LL = 63$ , which is bigger than the critical value of 5.99ish on  $df = 2$ ish

**Model 4c: Random Slope12, Random Slope26 Model**

Level 1:  $y_{ti} = \beta_{0i} + \beta_{1i}(\text{Slope12}_{ti}) + \beta_{2i}(\text{Slope26}_{ti}) + e_{ti}$   
 Level 2: Intercept:  $\beta_{0i} = \gamma_{00} + U_{0i}$   
 Slope12:  $\beta_{1i} = \gamma_{10} + U_{1i}$   
 Slope26:  $\beta_{2i} = \gamma_{20} + U_{2i}$

```
TITLE1 "Model 4c: Random Slope12, Random Slope26 Model";
PROC MIXED DATA=work.example34 NOCLPRINT NOITPRINT COVTEST NAMELEN=100 IC METHOD=REML;
  CLASS ID session;
  MODEL nm3rt = Slope12 Slope26 / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT Slope12 Slope26 / G GCORR V VCORR TYPE=UN SUBJECT=ID;
  REPEATED session / R TYPE=VC SUBJECT=ID;
  ESTIMATE "Session 1 Predicted Mean" Intercept 1 Slope12 0 Slope26 0;
  ESTIMATE "Session 2 Predicted Mean" Intercept 1 Slope12 1 Slope26 0;
  ESTIMATE "Session 3 Predicted Mean" Intercept 1 Slope12 1 Slope26 1;
  ESTIMATE "Session 4 Predicted Mean" Intercept 1 Slope12 1 Slope26 2;
  ESTIMATE "Session 5 Predicted Mean" Intercept 1 Slope12 1 Slope26 3;
  ESTIMATE "Session 6 Predicted Mean" Intercept 1 Slope12 1 Slope26 4;
  ESTIMATE "Difference between slopes" Slope12 -1 Slope26 1;
RUN; TITLE1;
```

```
TITLE "SPSS Model 4c: Random Slope12, Random Slope26 Model".
MIXED nm3rt BY ID session WITH Slope12 Slope26
  /METHOD = REML
  /PRINT = SOLUTION TESTCOV G R
  /FIXED = Slope12 Slope26
  /RANDOM = INTERCEPT Slope12 Slope26 | SUBJECT(ID) COVTYPE(UN)
  /REPEATED = session | SUBJECT(ID) COVTYPE(ID)
  /TEST = "Session 1 Predicted Mean" Intercept 1 Slope12 0 Slope26 0
  /TEST = "Session 2 Predicted Mean" Intercept 1 Slope12 1 Slope26 0
  /TEST = "Session 3 Predicted Mean" Intercept 1 Slope12 1 Slope26 1
  /TEST = "Session 4 Predicted Mean" Intercept 1 Slope12 1 Slope26 2
  /TEST = "Session 5 Predicted Mean" Intercept 1 Slope12 1 Slope26 3
  /TEST = "Session 6 Predicted Mean" Intercept 1 Slope12 1 Slope26 4
  /TEST = "Difference between slopes" Slope12 -1 Slope26 1.
```

```
* STATA Model 4c: Random Slope12, Random Slope26 Model
xtmixed nm3rt c.slope12 c.slope26, || id: slope12 slope26, ///
  variance reml covariance(un) residuals(independent,t(session)),
  estat ic, n(101),
  estat recovariance, level(id),
  estimates store RandP2,
  lrtest RandP2 RandP1
  lincom -1*c.slope12 + 1*c.slope26 // difference between slopes
```

**SAS output:**

Estimated R Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	17673					
2		17673				
3			17673			
4				17673		
5					17673	
6						17673

Estimated G Matrix

Row	Effect	Person ID	Col1	Col2	Col3
1	Intercept	101	284312	-54270	-10644
2	Slope12	101	-54270	63954	-1672.30
3	Slope26	101	-10644	-1672.30	2617.28

Estimated G Correlation Matrix

Row	Effect	Person ID	Col1	Col2	Col3
1	Intercept	101	1.0000	-0.4025	-0.3902
2	Slope12	101	-0.4025	1.0000	-0.1293
3	Slope26	101	-0.3902	-0.1293	1.0000

Estimated V Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	<b>301985</b>	230042	219399	208755	198111	187467
2	230042	<b>257400</b>	227410	215094	202778	190462
3	219399	227410	<b>235385</b>	208013	198314	188615
4	208755	215094	208013	<b>218604</b>	193850	186768
5	198111	202778	198314	193850	<b>207059</b>	184921
6	187467	190462	188615	186768	184921	<b>200747</b>

Estimated V Correlation Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.8251	0.8229	0.8125	0.7923	0.7614
2	0.8251	1.0000	0.9239	0.9068	0.8784	0.8379
3	0.8229	0.9239	1.0000	0.9170	0.8983	0.8677
4	0.8125	0.9068	0.9170	1.0000	0.9111	0.8916
5	0.7923	0.8784	0.8983	0.9111	1.0000	0.9070
6	0.7614	0.8379	0.8677	0.8916	0.9070	1.0000

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr >  Z
UN(1,1)	ID	284312	42731	6.65	<.0001
UN(2,1)	ID	-54270	18230	-2.98	0.0029
UN(2,2)	ID	63954	13244	4.83	<.0001
UN(3,1)	ID	-10644	3791.26	-2.81	0.0050
UN(3,2)	ID	-1672.30	2097.03	-0.80	0.4252
UN(3,3)	ID	2617.28	636.48	4.11	<.0001
Session	ID	17673	1435.84	12.31	<.0001

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8275.4	7	8289.4	8289.6	8296.8	8307.7	8314.7

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	1961.89	54.6805	100	35.88	<.0001
Slope12	-163.64	30.2188	100	-5.42	<.0001 RATE OF CHANGE FROM SESSION 1-2
Slope26	-32.8932	6.5888	100	-4.99	<.0001 RATE OF CHANGE FROM SESSION 1-6

Is random slope26 significant?  
 Yep,  $-2\Delta LL = 44$ , which is bigger than the critical value of 7.82ish on  $df \approx 3$ ish



Label	Estimates		DF	t Value	Pr >  t
	Estimate	Standard Error			
Session 1 Predicted Mean	1961.89	54.6805	100	35.88	<.0001
Session 2 Predicted Mean	1798.25	49.7847	100	36.12	<.0001
Session 3 Predicted Mean	1765.36	46.9899	100	37.57	<.0001
Session 4 Predicted Mean	1732.46	44.9935	100	38.50	<.0001
Session 5 Predicted Mean	1699.57	43.9044	100	38.71	<.0001
Session 6 Predicted Mean	1666.68	43.7905	100	38.06	<.0001
Difference between slopes	130.75	32.5530	100	4.02	0.0001

$$\text{Random Effect 95\% CI} = \text{fixed effect} \pm (1.96 * \sqrt{\text{Random Variance}})$$

$$\text{Intercept 95\% CI} = \gamma_{00} \pm (1.96 * \sqrt{\tau_{U_0}^2}) \rightarrow 1,961.9 \pm (1.96 * \sqrt{284,312}) = 917 \text{ to } 3,007$$

$$\text{Slope12 95\% CI} = \gamma_{10} \pm (1.96 * \sqrt{\tau_{U_1}^2}) \rightarrow -163.6 \pm (1.96 * \sqrt{63,954}) = -659 \text{ to } 322$$

$$\text{Slope26 95\% CI} = \gamma_{20} \pm (1.96 * \sqrt{\tau_{U_2}^2}) \rightarrow -32.9 \pm (1.96 * \sqrt{2,617}) = -133 \text{ to } 67$$

So far we've examined one way to fit piecewise slopes models—direct slopes that represent the change during each time period. Let's now examine an alternative specification: slope + deviation slope, which can be useful in examining individual differences in differential change between time periods.

**Model 5c: Random Slope, Random Deviation Slope Model**

Level 1:  $y_{it} = \beta_{0i} + \beta_{1i}(\text{Slope16}_{it}) + \beta_{2i}(\text{Slope26}_{it}) + e_{it}$   
 Level 2: Intercept:  $\beta_{0i} = \gamma_{00} + U_{0i}$   
           Slope12:  $\beta_{1i} = \gamma_{10} + U_{1i}$   
           Slope26:  $\beta_{2i} = \gamma_{20} + U_{2i}$

```
TITLE1 "Model 5c: Random Slope16, Random Slope26 Model";
PROC MIXED DATA=work.example34 NOCLPRINT NOITPRINT COVTEST NAMELEN=100 IC METHOD=REML;
  CLASS ID session;
  MODEL nm3rt = c1sess Slope26 / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT c1sess Slope26 / G GCORR V VCORR TYPE=UN SUBJECT=ID;
  REPEATED session / R TYPE=VC SUBJECT=ID;
  ESTIMATE "Slope between sessions 2-6" c1sess 1 Slope26 1;
RUN; TITLE1;

TITLE "SPSS Model 5c: Random Slope16, Random Slope26 Model".
MIXED nm3rt BY ID session WITH c1sess Slope26
  /METHOD = REML
  /PRINT = SOLUTION TESTCOV G R
  /FIXED = c1sess Slope26
  /RANDOM = INTERCEPT c1sess Slope26 | SUBJECT(ID) COVTYPE(UN)
  /REPEATED = session | SUBJECT(ID) COVTYPE(ID)
  /TEST = "Slope between sessions 2-6" c1sess 1 Slope26 1.

* STATA Model 5c: Random Slope16, Random Slope26 Model
xtmixed nm3rt c.c1sess c.slope26, || id: c1sess slope26, ///
  variance reml covariance(un) residuals(independent,t(session)),
  estat ic, n(101)
  estat recovariance, level(id)
  lincom 1*c.c1sess + 1*c.slope26 // slope between sessions 2 to 6
```

SAS output:

Estimated R Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	17673					
2		17673				
3			17673			
4				17673		
5					17673	
6						17673

Estimated G Matrix

Row	Effect	Person ID	Col1	Col2	Col3
1	Intercept	101	284312	-54270	43626
2	c1sess	101	-54270	63954	-65626
3	Slope26	101	43626	-65626	69916

Estimated G Correlation Matrix

Row	Effect	Person ID	Col1	Col2	Col3
1	Intercept	101	1.0000	-0.4025	0.3094
2	c1sess	101	-0.4025	1.0000	-0.9814
3	Slope26	101	0.3094	-0.9814	1.0000

Estimated V Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	<b>301985</b>	230042	219399	208755	198111	187467
2	230042	<b>257400</b>	227410	215094	202778	190462
3	219399	227410	<b>235385</b>	208013	198314	188615
4	208755	215094	208013	<b>218604</b>	193850	186768
5	198111	202778	198314	193850	<b>207059</b>	184921
6	187467	190462	188615	186768	184921	<b>200747</b>

Estimated V Correlation Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.8251	0.8229	0.8125	0.7923	0.7614
2	0.8251	1.0000	0.9239	0.9068	0.8784	0.8379
3	0.8229	0.9239	1.0000	0.9170	0.8983	0.8677
4	0.8125	0.9068	0.9170	1.0000	0.9111	0.8916
5	0.7923	0.8784	0.8983	0.9111	1.0000	0.9070
6	0.7614	0.8379	0.8677	0.8916	0.9070	1.0000

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr >  Z
UN(1,1)	ID	284312	42731	6.65	<.0001
UN(2,1)	ID	-54270	18230	-2.98	0.0029
UN(2,2)	ID	63954	13244	4.83	<.0001
UN(3,1)	ID	43626	19049	2.29	0.0220
UN(3,2)	ID	-65626	14154	-4.64	<.0001
UN(3,3)	ID	69916	15434	4.53	<.0001
Session	ID	17673	1435.84	12.31	<.0001

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8275.4	7	8289.4	8289.6	8296.8	8307.7	8314.7

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	1961.89	54.6805	100	35.88	<.0001
c1sess	-163.64	30.2188	100	-5.42	<.0001
Slope26	130.75	32.5530	100	4.02	0.0001

RATE OF CHANGE FROM SESSION 1-2  
DIFFERENCE IN RATE OF CHANGE FROM SESSION 2-6

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr >  t
Slope from session 2-6	-32.8932	6.5888	100	-4.99	<.0001


Random Effect 95% CI = fixed effect  $\pm (1.96 * \sqrt{\text{Random Variance}})$

Intercept 95% CI =  $\gamma_{00} \pm (1.96 * \sqrt{\tau_{U_0}^2}) \rightarrow 1,961.9 \pm (1.96 * \sqrt{284,312}) = 917 \text{ to } 3,007$

Slope16 95% CI =  $\gamma_{10} \pm (1.96 * \sqrt{\tau_{U_1}^2}) \rightarrow -163.6 \pm (1.96 * \sqrt{63,954}) = -659 \text{ to } 322$

Slope26 95% CI =  $\gamma_{20} \pm (1.96 * \sqrt{\tau_{U_2}^2}) \rightarrow 130.8 \pm (1.96 * \sqrt{69,916}) = -338 \text{ to } 649$

So how did we do? Let's compare model predictions in terms of means (top) and variances (bottom)?



See Hoffman chapter 6 for an example results section and complete description of these models, as well as a negative exponential model for these data.

